
FTP Push and SMS Alarms Using a Dual Interface 
Modem  
Data loggers are often used in locations which may be either inaccessible or inconvenient for 
manual downloading of data.  To remedy this, the dataTaker DT8x series data loggers have 
the ability to transfer data using networks and modems.  Advanced modems (such as the Call 
Direct CDM-882seu or CDR-790seu) are now widely available and can be used to initiate 
data transfer over the internet in addition to providing features such as dialling out and SMS 
message sending.   
 
The dataTaker can use its inbuilt schedules to power up an advanced modem, connect to the 
internet and upload recorded data to a remote FTP server and power down the modem.   
 

The benefits of this approach include: 

• simplifying the data retrieval process, 

• faster data transfer rates (than PTSN dial up), 

• offsite data storage (acts as a backup), 

• ability to access data without special/custom software needs, 

• password protection, 

• data can be downloaded anywhere in the world with an Internet connection, 

• SMS alarms can be sent through the same modem, and 

• lower power consumption by powering down the modem when not in use. 
 
The following sections in this technical note describe how to use the Call Direct CDM-882seu 
or CDR-790seu advanced modems to transfer data to an FTP server, send SMS messages 
and minimize power consumption. 

1 Prerequisites 
• Nil 

2 Equipment required 
• dataTaker DT80 range logger (firmware 8.00 or above) 

• Call Direct CDM-882seu/CDR-790seu dual-interface modem 

• UTMS/HSDPA compatible SIM card  

• Cross-over Ethernet cable 

• Serial RS232 DE-9 cable 

• 5.5mm OD / 2.1mm ID DC Jack and suitable cable 
 
NOTE: The SIM card must be configured for data access by the network carrier.  You may 
need to ask the carrier to enable ‘Internet access’. 

3 Modem set up 
NOTE: If you require additional information regarding the other functions or advanced 
settings for the Call Direct CDM-882seu or CDR-790seu modem, please consult the 
user manual, which can be freely obtained from www.call-direct.com.au  
 
Before the modem can be used with the dataTaker, it needs to be configured.  In order to do 
this, first insert your SIM card, power up the modem and plug it into the Ethernet port of your 
PC using a cross-over Ethernet cable. Ensure that you have automatic IP addressing on your 
PC (which you can check from control panel -> network connections), otherwise further set 
up will be required.  Once your computer has established a connection to the modem, open 
your web browser and type in the address “192.168.1.50” into the address bar and press 
enter. 

The default username will be admin and the password will be password.  Enter these details 
when prompted and you will be directed to the status page: 

 
 
Each item in the left hand column relates to a configuration page, but most of them can be left 
as default and only the following pages need to be configured: 

• Data Connection 

• Modem 

• Security 

3.1 Data Connection page 
 

 
 

• Change Auto connect to Enable 

• Click Save 

• The PPP User Settings may be stored on the SIM card and are network specific.  If 
the settings are blank or incorrect, you will need to contact your network carrier for 
assistance. An example of the APN and dialing number is ‘telstra.internet’ and 
‘atd*99#’ as shown above. 

• Some networks require a username and password; these should be provided by the 
network carrier. 

3.2 Modem page 

 
• Change Inter Character Timeout to 1000 

• Change RI Action to Always Off 

• Change Circuit AutoAnswer Rings to Off 

• Click Save 

3.3 Security page  
This page only needs to be configured if the SIM card is locked and has a PIN, if so then you 
may enter the details here. 

3.4 Back up your new configuration 

Go to the Application Load/Save Page and click the SAVE button to back up these settings 
so that they can be restored in the future.  

4 Connecting the modem to the dataTaker 
There are two points where the modem connects to the dataTaker for communications, more 
so when the dataTaker is in control of the power to the modem. 

4.1 Ethernet 
Connect the two devices using a cross-over Ethernet cable.  Ensure the cable is secure and 
that the two ends ‘click’ into place.  

4.2 Serial Cable 
Connect the two devices using a DB9 serial cable that has the (RS232 DE-9) pinout: 

 
 

Pin Name Direction Description 

1 CD � Carrier Detect 

2 RX � Receive Data 

3 TX  Transmit Data 

4 DTR  Data Terminal Ready 

5 GND  System Ground 

6 DSR � Data Set Ready 

7 RTS  Request to Send 

8 CTS � Clear to Send 

9 RI � Ring Indicator 

 
Use the screws to ensure that the cable is secured and will not fall out.  

4.3 Power 
This section only applies to those who have power sensitive applications and where the 
dataTaker is to control the power of the modem.  In this case the switching RELAY terminals 
on the logger are controlled to connect and disconnect the power to the modem.  The power 
source can be from a solar conditioner, external 12V battery, AC power pack or otherwise, but 
not the internal battery.  It is not recommended to power a modem using the 12V output 
terminal because it is current limited to 150mA.  To use the RELAY, follow these steps: 

• Obtain a 5.5mm/2.1mm DC jack with cable (Figure 1) such as those available from 
http://www.action-electronics.com/dcpowerplugs.htm 

 

 

Figure 1 - DC Jack with cable 

• Use a multi meter and check which of the two wires connects to the pin (inner 
terminal) of the DC Jack, this is the positive wire. 

• Cut the positive wire (identified above) approximately 20cm from the end 

• Strip the wire insulation back about 5mm in each direction of the cut to provide two 
bare-wire connections. (You may tin this with a soldering iron if desired) 

• Secure these two bare-wire ends into the RELAY A and B inputs on the dataTaker 
(as shown in Figure 2). 

 

 

Figure 2 - Relay connection for the modem power cable 

• Connect the positive wire to the power supply positive terminal, which is on the side 
of the logger 

• Connect the negative wire into the negative power supply terminal on the side of the 
logger 

• Double-check your connections and that they are secure 

• Turn on the power source 

• Check with your multi meter that there is no voltage supply to the modem DC jack 

• Issue the 1RELAY=1 command to the dataTaker using the web interface or 
DeTransfer 

• Re-measure the voltage across the DC jack terminals, ensure that the pin (inner 
terminal) has a positive voltage with respect to the outer, and that the voltage is that 
of your power source (8-28V source dependant). 

• Now plug the newly wired DC jack into the modem and check that it powers up. 
 

You are now ready to set up the dataTaker. 

5 dataTaker configuration 
Now that the modem has been set up, the dataTaker can now be configured to talk to the 
modem. 

5.1 Profile settings 
Using either the web interface or DeTransfer, issue the following profile commands to the 
logger. 

5.1.1 Ethernet settings 

The IP address has to be specified in order to avoid conflicts and any DHCP delays.  The 
gateway is the IP address of the modem (which is 192.168.1.50 by default). 
 

PROFILE ETHERNET IP_ADDRESS="192.168.1.51" 

      PROFILE ETHERNET SUBNET_MASK="255.255.255.0" 

      PROFILE ETHERNET GATEWAY="192.168.1.50" 

5.1.2 Host-modem settings 

The host-modem strings have been cleared because this application note does not cover 
dialing out with the CDM-882seu  
 

      PROFILE HOST_MODEM DIAL="" 

      PROFILE HOST_MODEM INIT="" 

5.1.3 Host-port settings 

The host-port settings have to match those specified by the CDM-882seu set up pages, 
otherwise you will not be able to communicate to the modem.  The default settings for the 
modem are as follows: 
 

      PROFILE HOST_PORT BPS="115200" 

      PROFILE HOST_PORT FLOW="NONE" 

      PROFILE HOST_PORT DATA_BITS="8" 

      PROFILE HOST_PORT STOP_BITS="1" 

      PROFILE HOST_PORT PARITY="NONE" 

      PROFILE HOST_PORT FUNCTION="SERIAL" 

5.2 dataTaker program 1 (FTP push only) 

5.2.1 Overview 

The following describes a basic dataTaker program which uses the Call Direct CDM-882seu 
modem to push data to an FTP server.  Functionality built into this program includes 
automatic retries, but it does not include control of the power to the modem, or provision for 
SMS sending.  It is also assumed that the power to the modem is not controlled by the 
dataTaker, and the modems data connection is always active. 
 
The K (FTP handler) schedule handles all FTP transfers.  It uses the COPYD command 
which keeps track of the data that has successfully been copied in the past, therefore it will 
only transfer new data to the FTP server.  On a failed attempt it will recall the point at which 
the copy was first attempted and copy this plus any new data on subsequent attempts.  

5.2.2 Program Code 
BEGIN "CD882B1" 

'==================================================== 

'Call Direct CDM882seu Basic Sample Program number 1 

'   - Sends all logged data to an FTP server each hour  

'==================================================== 

  

'==================================================== 

' Schedule A (SAMPLE USER SCHEDULE) 

'   - Logs up to 1MB of data records and overwrites when full 

'   - Runs every 10 seconds 

'==================================================== 

RA(DATA:OV:1MB)10S LOGONA 

     REFT("Internal_Temp") 'sample internal temperature 

     1-TK("External_Temp") 'sample external temperature 

'==================================================== 

' Schedule K (FTP HANDLER) 

'   - Copies new data to an ftp server every hour  

'   - If a transfer fails, a transfer retry will occur next hour 

'   - ALL schedules will be copied, unless specified (i.e. sched=ab) 

'   - For polled pushes, change to a polled schedule (RKX) 

'==================================================== 

RK1H LOGOFFK 

    DO{COPYD merge=N id=1 start=new dest=ftp://user:pass@host.com/}  

END 

5.3 dataTaker program 2 (basic SMS program) 

5.3.1 Overview 

The following describes a very basic dataTaker program which uses the Call Direct CDM-
882seu modem to send SMS messages.  Functionality built into this program does not 
include control of the power to the modem, provision for FTP transfers or any smart control of 
the SMS queue.  It is also assumed that the power to the modem is not controlled by the 
dataTaker, and the modem is connected to a mobile network. 

5.3.2 How it works 

The SMS command is sent to the modem as a set of serial AT commands which follows this 
sequence (from the logger perspective): 
 
A). � AT+CMGF=1<CR>     (send AT command to force text messaging mode) 
B).  Logger waits 500ms for command to take effect 
C). � AT+CMGS="+61412345678"<CR>   (request to send to specified number) 
D).  Logger waits for prompt ">" character 
E). � MESSAGE<ESC><CR> (send text message contents) 
F).  Logger waits 5 seconds for message to be sent (forced delay from the modem) 
 

5.3.3 Program Code 
BEGIN "CD882B2" 

'==================================================== 

'Call Direct CDM882seu Basic Sample Program number 2 

'   - Sends SMS messages only  

'==================================================== 

 

'==================================================== 

' Schedule A (SAMPLE USER SCHEDULE) 

'   - Runs every 10 seconds 

'==================================================== 

RA("B:",ALARMS:OV:10KB,DATA:OV:1MB)10S LOGONA 

    'measure the sensors  

    1TK("Sensor1") 

    'alarms to send SMS message in text format using AT+CMGS command 

    ALARM(&Sensor1>30){2SERIAL("{AT+CMGF=1\\013}\\W[500]")}  

    ALARM(&Sensor1>30){2SERIAL("{AT+CMGS=\\034}")} 'send SMS command 

    ALARM(&Sensor1>30)'+61412345678'     'enter the phone number here 

    'the following line instructs to wait for the ">" character 

    ALARM(&Sensor1>30){2SERIAL("{\\034\\013}\\m[\\062]")}  

    ALARM(&Sensor1>30)'MESSAGE'  'enter the message text here 

    ALARM(&Sensor1>30){2SERIAL("{\\026\\013}\\W[5000]")}'delay 

END 

5.4 dataTaker program 3 (multiple SMS alarms) 

5.4.1 Overview 

The following describes a dataTaker program which uses the Call Direct CDM-882seu 
modem to handle multiple alarms and SMS messages.  Functionality built into this program 
does not include control of the power to the modem, or provision for FTP transfers.  It is also 
assumed that the power to the modem is not controlled by the dataTaker, and the modem is 
connected to a mobile network. 
 
This program has the capability to send multiple alarms, which are queued to be sent 
sequentially.  Often modems require a delay between sending SMS messages and this 
program caters for this by using a 5-second schedule, which will send one message from the 
queue each time it is run.  It is possible to simplify this program by using the DELAY 
command; however this would delay operation of all schedules and therefore is not ideal. 

5.4.2 A Schedule 

The A schedule is a sample of how to implement the alarms, it contains two measurements 
and three alarm lines.  The key to this schedules operation is a queue for each alarm (201CV, 
202CV, etc) and a main queue (200CV), which are incremented when the alarm is true.   
 
NOTE: Avoid using repeating alarms (ALARMR), otherwise many SMSes will be put into one 
queue and prevent other alarms from being sent. 

5.4.3 K Schedule 

The K schedule runs every 5 seconds, interrogating the global queue.  If the number of 
messages in the global queue is greater than zero then it will progress through the individual 
SMS queues one at a time, send any outstanding SMS alarms and adjust the queue 
accordingly.  The SMS commands are sent in the same order as per section 5.3.2. 
 
NOTE 1: Each polled alarm must have an IF statement in the following format (where the 
orange text is the message. 

 
    IF(200CV>1)AND IF(201CV>1)'Alarm 1'{201CV=201CV-1}  

 
In the event that there is more than one alarm used in the program then priority becomes 
important, and the messages must be sent in order.  In this case, the previous queue must 
also be empty therefore an entry would look like this: 
 

    IF(201CV<1)AND  

    IF(200CV>1)AND IF(202CV>1)'Alarm 2'{202CV=202CV-1} 

 
Putting it together, it would look like this: 
 
    IF(202CV<1)AND 

    IF(200CV>1)AND IF(203CV>1)'Alarm 3'{203CV=203CV-1} 

 

    IF(201CV<1)AND  

    IF(200CV>1)AND IF(202CV>1)'Alarm 2'{202CV=202CV-1} 

 

    IF(200CV>1)AND IF(201CV>1)'Alarm 1'{201CV=201CV-1}  

 
Notice the reverse order and that “Alarm 2” requires the “Alarm 1” queue (201CV) to be 
empty, and “Alarm 3” requires the “Alarm 2” queue (202CV) to be empty. 

5.4.4 Program Code 
BEGIN "CD882B3" 

'==================================================== 

'Call Direct CDM882seu Basic Sample Program number 3 

'   - Sends SMS messages based on different alarms   

'   - Each alarm message has an associated queue, and one global 

'   - Two queues are included which can be expanded upon  

'   - SMS phone number and messages are defined in the K schedule 

'==================================================== 

 

'==================================================== 

'Initialise variables 

    200..203CV("GlobalQueue",W)=0 'reset queues 

'==================================================== 

 

'==================================================== 

' Schedule A (SAMPLE USER SCHEDULE) 

'   - Runs every 10 seconds 

'==================================================== 

RA("B:",ALARMS:OV:10KB,DATA:OV:1MB)10S LOGONA 

    'measure the sensors  

    1TK("Sensor1") 

    2TK("Sensor2") 

    'TO SEND AN SMS ALARM, Increment the global and individual SMS 

queues 

    ALARM(&Sensor1>30){200CV=200CV+1 201CV=201CV+1}'queue1 

    ALARM(&Sensor2<01){200CV=200CV+1 202CV=202CV+1}'queue2 

    ALARM(&Sensor2>40){200CV=200CV+1 203CV=203CV+1}'queue3 

 

'==================================================== 

' Schedule K (SMS HANDLER SCHEDULE) 

'   - Processes the SMS queues one by one every 5 seconds 

'==================================================== 

RK5S LOGOFFK  

    'if a message in the queue, start the message sequence 

  IF(200CV>1){2SERIAL("{AT+CMGF=1\\013}\\w[500]{AT+CMGS=\\034}")} 

    IF(200CV>1)'+61412345678'   'enter the phone number here 

    IF(200CV>1){2SERIAL("{\\034\\013}\\m[\\062]")} 'wait for ">" 

    'send messages in order of priority (queue1 = highest priority) 

    'The order is reversed programmatically (203CV->201CV) 

    IF(202CV<1)AND 

    IF(200CV>1)AND IF(203CV>1)'Sensor3>40 deg'{203CV=203CV-1} 

    IF(201CV<1)AND  

    IF(200CV>1)AND IF(202CV>1)'Sensor2<01 deg'{202CV=202CV-1} 

    IF(200CV>1)AND IF(201CV>1)'Sensor1>30 deg'{201CV=201CV-1}  

    'complete the serial string and remove 1 from the global queue 

    IF(200CV>1){2SERIAL("{\\026\\013}") 200CV=200CV-1}   

END 

5.5 Advanced dataTaker program 

5.5.1 Overview 

The following program incorporates four schedules which are dedicated to the operation of 
the modem (schedules H I J & K) and two generic schedules (A & B) which are user 
configurable. There are two queue systems used in this program, an SMS queue (as per 
section 5.4) and an FTP queue (104CV).  In addition, the logger detects the modem 
connection to the mobile data network by monitoring the state of the modem hardware lines.  
Using these functionalities it is possible for the dataTaker to power-up the modem, send 
SMSes and FTP data, then power-down then modem to save power. 

5.5.2 The A Schedule 

The A schedule takes measurements from a digital input and a thermocouple and tests these 
against alarm conditions.  If the alarm condition is met, the SMS queues are incremented (see 
section 5.4 for details) and the modem power handler.  The alarm will not re-trigger unless the 
condition is false, then true again, this prevents SMSes being sent continuously.  There are 
four parameters which can be changed for this schedule, including the schedule rate, 
schedule letter, measurement types and alarm conditions. 
Note the SMS message itself and its destination phone number is defined in the J schedule. 

5.5.3 The B Schedule 

The B (FTP) schedule queues an FTP request once per hour and calls the modem power 
handler. The schedule rate determines the time period for the FTP data.  The FTP server 
address is defined in schedule K. 

5.5.4 The H Schedule 

The H (Modem power handler) schedule, when polled, provides power to the modem via a 
relay. When the relay is closed, the ATTN light on the dataTaker front panel is activated.  If 
the modem is already powered up, polling the H-schedule has no effect. 

5.5.5 The I Schedule 

The I (Modem connection handler) schedule waits for the modem to assert the DCD, DSR 
and CTS lines on the RS232 connection, which indicates that the modem is online and 
connected to the data network.  It retries failed connection attempts by cycling the modem 
power.  Once online, it activates the J schedule. 

5.5.6 The J Schedule 

The J (online activity handler) schedule transfers the SMS messages and FTP data.  The 
sending of SMS messages is as per section 5.4, in parallel with the FTP transfers.  The 
system variable 29SV is monitored to determine the status of the FTP transfer.  Once the 
transfer is complete or has failed for any reason, the transfer is removed from the queue and 
the hang-up schedule is started.  In the event that an FTP transfer has failed, the data will be 
sent on subsequent attempts. The only parameters that can be changed here are the SMS 
message and phone number and the FTP destination; otherwise this should not be edited. 

5.5.7 The K Schedule 

The K (Modem hang-up handler) schedule waits for all FTP transfers to complete and and all 
SMS messages to be sent, then hangs up the modem. 

5.5.8 Program Code 
BEGIN "CD882A1" 

'==================================================== 

'Call Direct CDM882seu Advanced Sample Program  

'   - Controls modem power to save total power consumption 

'   - Sends FTP data for all schedules 

'   - SMS queues are supported  

'   - SMS phone number and messages are defined in the K schedule 

'==================================================== 

 

'==================================================== 

'Initialise variables 

    104CV("FTP QUEUE",W)=0 

    200..202CV("SMS QUEUES",W)=0 

    '110CV are reserved 

    1RELAY=0 1WARN=0 'modem and warning light off 

'==================================================== 

 

'==================================================== 

'USER SETTINGS 

    106CV("CONNECT_TIMEOUT",W)=12 'x10 seconds 

    109CV("CONNECT_RETRIES",W)=4 'number of modem power cycles to attempt 

'==================================================== 

 

'==================================================== 

'Schedule A: Sample User Schedule 

'   - Measures the current load and monitors the load threshold 

'   - Logging ON (2MB for data, 10KB for alarms) 

'   - Runs every 10 seconds 

'  TO SEND AN SMS ALARM: 

'   - Increment the global (200CV) and individual (201CV, 202CV, etc) SMS queues 

'   - Call the H schedule  

'==================================================== 

RA(DATA:2MB,ALARMS:10KB)10S LOGONA  

    'take measurements 

    4DS("Measurement1",NR) 

    1TK("Temperature1",NR) 

    'set alarms 

    ALARM1(&Measurement1<01)"ALARM 1 TRIGGERED"{200CV=200CV+1 201CV=201CV+1 XH}  

    ALARM2(&Temperature1>30)"ALARM 2 TRIGGERED"{200CV=200CV+1 202CV=202CV+1 XH} 

'==================================================== 

'Schedule B: Sample FTP Scheduler 

'   - Triggers data to be sent to an FTP server every hour 

'   - Logging ON (10KB for alarms) 

'==================================================== 

RB(ALARMS:10KB)1H LOGONB 

    104CV(W)=1  'schedule FTP transfer 

    DO"FTP TRIGGERED"{XH}      'call the modem power handler 

'==================================================== 

'Schedule H: Modem power handler ===DO NOT EDIT=== 

'   - Call this when a transmission is required 

'   - Checks if the modem is connected to the network 

'     - If not, powers up the modem and attempts connection 

'   - Logging ON (10KB for alarms) 

'   - 16SV Monitors hardware lines to detect DCD, DSR and CTS signals  

'====================================================



©2010 Thermo Fisher Scientific Australia Pty Ltd. All rights reserved. A.B.N. 52 058 390 917

For customer service, call 1300-735-292
To fax an order, use 1800-067-639
Visit us online: www.thermofisher.com.au

  

 Technical Note 
 

 TN-0031 

  

TN-0031-B1 

RH(ALARMS:10KB)X LOGONH 

    IF(16SV<1)"Powering ON modem"{1WARN=1 1RELAY=1 110CV=109CV 107CV=106CV+1 GI} 

'==================================================== 

'Schedule I: Modem connection handler ==DO NOT EDIT== 

'   - Waits for the modem to connect to the mobile network 

'   - Cycles the power on the modem on connection failure 

'   - Logging ON (10KB for alarms) 

'   - Activates online functions once connected 

'====================================================   

RI(ALARMS:10KB)10S HI LOGONI 

    107CV("Retries remaining",W)=107CV-1      

    'if powered down (due to cycling power), power back on  

    IF(1RELAY<1)"Modem power re-applied,(?110f0) retries remaining "{1RELAY=1} 

    'Connection timeout. Power down modem and decrement retries remaining 

    IF(107CV<1)"Resetting Modem"{1RELAY=0 107CV=106CV+1 110CV(W)=110CV-1} 

    IF(110CV<1)"Cannot connect to network"{HI} 

    IF(16SV>7)"Connected"{101CV=0 GJ HI}   

'==================================================== 

'Schedule J: Online Activity Handler ==DO NOT EDIT== 

'   - Attempts to send any FTP data and SMS messages  

'   - SMS messages processed according to priority 

'   - runs every 10 seconds until modem hangs up 

'   - Logging ON (100KB for alarms) 

'====================================================  

RJ(ALARMS:100KB)10S  HJ LOGONJ 

    'Test FTP status before attempting transfer,101CV provides 10 second delay  

    IF(101CV>1)AND IF(29SV>2)"FTP Transfer success at # @"{104CV=0 101CV=0} 

    IF(101CV>1)AND IF(29SV<0)"FTP Transfer failed at # @"{104CV=0 101CV=0}  

    'initiate FTP transfer for all new data in all schedules 

    IF(104CV>1)AND  

    IF(29SV<>1,2){COPYD dest=ftp://user:pass@ip_address/ merge=N id=1 start=new}   

    'send SMS messages in the queue 

    IF(200CV>1){2SERIAL("{AT+CMGF=1\\013}\\w[500]{AT+CMGS=\\034}")} 

    IF(200CV>1)'+61412345678'   'enter the phone number here 

    IF(200CV>1){2SERIAL("{\\034\\013}\\m[\\062]")} 'wait for ">" 

    IF(201CV<1)AND 

    IF(200CV>1)AND IF(202CV>1)'Temperature 1 >30 deg'{202CV=202CV-1} 

    IF(200CV>1)AND IF(201CV>1)'Measurement 1 OFF'{201CV=201CV-1}  

    IF(200CV>1)"SMS SENT"{2SERIAL("{\\026\\013}\\m[CMGS]") 200CV=200CV-1}   

    IF(200CV>1)"?200f0 SMS still in Queue" 

    101CV("SecondRun",W)=1  'set flag  

    DO{GK}   'start hang-up handler  

'==================================================== 

'Schedule K: Modem Hang-up Handler ==DO NOT EDIT== 

'   - Runs every 1 second  

'   - Hangs up modem when all activities are completed 

'   - Logging ON (1KB for alarms) 

'====================================================  

RK(ALARMS:1KB)1S  HK LOGONK 

    'If FTP and SMS queues empty, reset connect flag and turn off modem 

    IF(104CV<1)AND IF(200CV<1)"Powering OFF modem"{1RELAY=0 1WARN=0 HJ HK} 

END 

 
 


